Name: KEY **Id#**

COE 405, Term 062

Design & Modeling of Digital Systems

Quiz# 3

Date: Saturday, April 21, 2007

Q.1.

(i) Your are required to model the function "-" to compute the 2's complement of a Bit_Vector operand. You are required to model the function without the use of addition and subtraction operation on any type. The function should be modeled to be general enough to handle inputs specified using any range format i.e. To or Downto format. Define the function inside a package and use the package whenever the function needs to be used.

```
Package Quiz3Pack is
Function "-" (l: IN Bit_Vector) Return Bit_Vector;
End;
Package Body Quiz3Pack is
Function "-" (l: IN Bit_Vector) Return Bit_Vector IS
Alias tmp1: Bit_Vector(l'length-1 Downto 0) Is l;
Variable tmp2: Bit_Vector(l'length-1 Downto 0);
Variable Found1: Boolean;
Begin
   For i in 0 to l'length-1 Loop
         If (Not Found1) Then
               If tmp1(i) = '1' Then
                     Found1 := True;
               End If;
               tmp2(i) := tmp1(i);
         else
               tmp2(i) := not tmp1(i);
         End if;
   End Loop;
   Return tmp2;
End "-";
End;
```

(ii) Model an entity ABSV to return the absolute value of a Bit_Vector type using this function. Assume that the input represented in Bit_Vector is a signed number in 2's complement format. The entity should be parametrizable with a generic parameter N. You are free to use any modeling style for the architecture.

```
use work.Quiz3Pack.all;
Entity ABSV is
Generic (N: Natural:=8);
Port (X: IN Bit_Vector(N-1 Downto 0); Y: OUT Bit_Vector(N-1 Downto 0));
End;
Architecture Quiz3 of ABSV is
Begin
    Y <= X when X(N-1)='0' Else -X;
End;
```